#region license
// Copyright (c) 2007 Rodrigo B. de Oliveira (rbo@acm.org)
// All rights reserved.
// 
// Redistribution and use in source and binary forms, with or without modification,
// are permitted provided that the following conditions are met:
// 
//     * Redistributions of source code must retain the above copyright notice,
//     this list of conditions and the following disclaimer.
//     * Redistributions in binary form must reproduce the above copyright notice,
//     this list of conditions and the following disclaimer in the documentation
//     and/or other materials provided with the distribution.
//     * Neither the name of Rodrigo B. de Oliveira nor the names of its
//     contributors may be used to endorse or promote products derived from this
//     software without specific prior written permission.
// 
// THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND
// ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED
// WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE
// DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT OWNER OR CONTRIBUTORS BE LIABLE
// FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL
// DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR
// SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER
// CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY,
// OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF
// THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.

# Taken and modified from ConcurrentBoo

# This file is part of Grawlix.
# 
# Grawlix is free software: you can redistribute it and/or modify
# it under the terms of the GNU General Public License as
# published by the Free Software Foundation, either version 3 of
# the License, or (at your option) any later version.
# 
# Grawlix is distributed in the hope that it will be useful,
# but WITHOUT ANY WARRANTY; without even the implied warranty of
# MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
# GNU General Public License for more details.
# 
# You should have received a copy of the GNU General Public
# License along with Grawlix.  If not, see
# <http://www.gnu.org/licenses/>.
#endregion

"""Primitive operations."""

namespace ConcurrentBoo
import ConcurrentBoo.Executors
import System.Threading

static class Primitive:
"""Primitive concurrent operations."""

	callable Action()
	"""A callable operation with no return value."""
	
	callable Expression() as object
	"""A callable operation with a return value."""
	
	def Asynchronously([required] action as Action):
	"""
	Run an action in the ThreadPool, saving the result.
	Param action: The action to invoke.
	Returns: A computation object holding the result.
	"""
		computation = Computation()
		QueueUserWorkItem() do:
			Invoke(computation, action)
		return computation
	
	def Asynchronously([required] executor as IExecutor,
	           [required] action as Action):
	"""
	Run an action using an IExecutor, saving the result.
	Param executor: The IExecutor to run the action.
	Param action: The action to invoke.
	Returns: A computation object holding the result.
	"""
		computation = Computation()
		executor.Execute() do:
			Invoke(computation, action)
		return computation
	
	def Future([required] expression as Expression):
	"""
	Run an expression in the ThreadPool, saving the result.
	Param expression: The expression to invoke.
	Returns: A computation object holding the result.
	"""
		computation = Computation()
		QueueUserWorkItem() do:
			Invoke(computation, expression)
		return computation
	
	def Future([required] executor as IExecutor,
	           [required] expression as Expression):
	"""
	Run an expression using an IExecutor, saving the result.
	Param executor: The IExecutor to run the expression.
	Param action: The expression to invoke.
	Returns: A computation object holding the result.
	"""
		computation = Computation()
		executor.Execute() do:
			Invoke(computation, expression)
		return computation
	
	def Invoke([required] action as Action):
	"""
	Run an action, saving the result in a new computation object.
	Param action: The action to evaluate.
	Returns: The new computation.
	"""
		return Invoke(Computation(), action)
	
	def Invoke([required] computation as Computation,
	           [required] action as Action):
	"""
	Run an action, saving the result in a new computation object.
	Param computation: The computation to store the result.
	Param action: The action to evaluate.
	Returns: The computation.
	"""
		try:
			action()
			computation.Value = null
		except exception:
			computation.Failure = exception
		return computation
	
	def Invoke([required] expression as Expression):
	"""
	Run an expression, saving the result in a new computation object.
	Param expression: The expression to evaluate.
	Returns: The new computation.
	"""
		return Invoke(Computation(), expression)
	
	def Invoke([required] computation as Computation,
	           [required] expression as Expression):
	"""
	Run an expression, saving the result in a computation object.
	Param computation: The computation to store the result.
	Param expression: The expression to evaluate.
	Returns: The computation.
	"""
		try:
			computation.Value = expression()
		except exception:
			computation.Failure = exception
		return computation
	
	def QueueUserWorkItem([required] callback as WaitCallback):
	"""
	Shorthand for ThreadPool.QueueUserWorkItem.
	Param callback: The operation to queue.
	"""
		assert ThreadPool.QueueUserWorkItem(callback)
	
